Politecnico di Torino

22-29 March 2017

# Operations Research: theory and applications to networking

*Lab 1- The Bin Packing Problem*

**Computer and Communication Engineering**

# Lab 1 (not included in the final report)

Formulate the bin packing problem as an optimization problem

Discuss all the variables introduced and the type of problem

Solve the problem with XpressMP assuming: 8 candidate bins, 8 items of size {4,5,6,6,8,9,10,11}, Maximum Bin Capacity = 20

Print the values of all the variables of the problem.

Solve again the problem for the following cases

- bin capacity set to 11
- bin capacity set to 1

What happens to the solution in both cases?

# Problem Definition

**We have a set of items of different size**



**We want to store items in bins with limited capacity**



# Problem Statement

**Minimize**

The number of used bins

**Subject to**

- Each item is assigned to one bin
- Maximum bin capacity

**With Variables**

Bin Used {Yes,No}
Item Assigned to Bin {Yes,No}

# Parameters and Variables

## Parameters (input, fixed)

$$Q_j : \text{capacity of bin } j$$

$$q_i : \text{size of item } i$$

## Variables (output)

$$x_{ij} = \begin{cases} 1, & \text{if item } i \text{ in bin } j \\ 0, & \text{otherwise} \end{cases}$$

$$y_j = \begin{cases} 1, & \text{if bin } j \text{ is used} \\ 0, & \text{otherwise} \end{cases}$$

# Mathematical Formulation

$$\text{Minimise} \quad \sum_{j=1}^{n} y_j$$

Minimize the number of used bins

$$s.t. \quad x_{ij} \le y_j, \quad \forall i, j$$

A bin is used if at least one item is assigned to it

$$\sum_{j=1}^{n} x_{ij} = 1, \quad \forall i$$

An item can be assigned to only one bin

$$\sum_{j=1}^{n} q_i x_{ij} \le Q_j, \quad \forall j$$

The sum of items size assigned to the bin must be lower than the bin capacity

$$x_{ij} \in \{0,1\}, \quad \forall i, j$$

$$y_j \in \{0,1\}, \quad \forall j$$

Binary Variables

# Mosel Translation

## Declarations and Parameters

```
model "BinPacking"

    uses "mmxprs"    !Use Xpress-Optimiser

    declarations     !Parameters declaration
        BIN = 1..8   !Number of bins
        ITEM = 1..8  !Number of items

        y: array(BIN) of mpvar        !Decision variable for bin
        x: array(ITEM,BIN) of mpvar   !Decision variable for item

        BINSIZE: array(BIN) of integer        !Input Parameter: Bin Size
        ITEMSIZE: array(ITEM) of integer      !Input Parameter: Item Size
    end-declarations

    !Initialise Bin Size and Item Size
    BINSIZE::  [20,20,20,20,20,20,20,20]
    ITEMSIZE::  [4,5,6,6,8,9,10,11]

```

# Mosel Translation - II

## Model Formulation

$$\sum_{j=1}^{n} y_j$$

$$x_{ij} \le y_j, \quad \forall i, j$$

$$\sum_{j=1}^{n} x_{ij} = 1, \quad \forall i$$

$$\sum_{j=1}^{n} q_i x_{ij} \le Q_j, \quad \forall j$$

```
!Objective Function
NumBin:= sum(j in BIN) y(j)

!Constraints
forall(i in ITEM, j in BIN) x(i,j) <= y(j)
forall(i in ITEM) sum(j in BIN) x(i,j) = 1
forall(j in BIN) sum(i in ITEM) x(i,j)*ITEMSIZE(i) <= BINSIZE(j)

!Binary variables
forall(i in BIN) y(i) is_binary
forall(i in ITEM,j in BIN) x(i,j) is_binary

!Optimisation statement
minimize(NumBin)
```

$$y_j \in \{0,1\}, \quad \forall j$$

$$x_{ij} \in \{0,1\}, \quad \forall i, j$$

*Minimise*

# Mosel Translation - III

## Output Display

```
!Display output
writeln("Number of bins used: ", getobjval)

forall(j in BIN) do
    if getsol(y(j)) = 1 then
        writeln("Bin ", j, " is used. (Size: ", BINSIZE(j), ")")
        writeln(" Items: ")

        forall(i in ITEM)
            if getsol(x(i,j)) = 1 then
                writeln(" ", i, " (Size: ", ITEMSIZE(i), ")")
            end-if
    else
        writeln("Bin ", j, " is NOT used. (Size: ", BINSIZE(j), ")")
        writeln(" Items: NONE.")
    end-if
end-do

end-model
```

Time to work!